C06DBF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

C06DBF returns the value of the sum of a Chebyshev series through the routine name.

2 Specification

```
real FUNCTION C06DBF(X, C, N, S)
INTEGER              N, S
real                 X, C(N)
```

3 Description

This routine evaluates the sum of a Chebyshev series of one of three forms according to the value of the parameter S:

$$S = 1: \quad 0.5c_1 + \sum_{j=2}^{n} c_j T_{j-1}(x),$$

$$S = 2: \quad 0.5c_1 + \sum_{j=2}^{n} c_j T_{2j-2}(x),$$

$$S = 3: \quad \sum_{j=1}^{n} c_j T_{2j-1}(x)$$

where x lies in the range $-1.0 \leq x \leq 1.0$. Here $T_r(x)$ is the Chebyshev polynomial of order r in x, defined by $\cos(ry)$ where $\cos y = x$.

The method used is based upon a three-term recurrence relation; for details see Clenshaw [1].

4 References

[1] Clenshaw C W (1962) Mathematical tables *Chebyshev Series for Mathematical Functions* HMSO

5 Parameters

1: X — ***real*** *Input*

On entry: the argument x of the series.

Constraint: $-1.0 \leq \mathrm{X} \leq 1.0$.

2: C(N) — ***real*** array *Input*

On entry: C(j) must contain the coefficient c_j of the Chebyshev series, for $j = 1, 2, \ldots, n$.

3: N — INTEGER *Input*

On entry: the number of terms, n, in the series.

4: S — INTEGER *Input*

On entry: S must have the value 1, 2 or 3 according to whether the series is general, even or odd respectively (see Section 3). For all other values of S, the routine behaves as though S = 2.

6 Error Indicators and Warnings

None.

7 Accuracy

There may be a loss of significant figures due to cancellation between terms. However, provided that n is not too large, the routine yields results which differ little from the best attainable for a given word length.

8 Further Comments

The time taken by the routine increases with n.

This routine has been prepared in the present form to complement a number of integral equation solving routines which use Chebyshev series methods, e.g., D05AAF and D05ABF.

9 Example

This program evaluates

$$0.5 + T_1(x) + 0.5T_2(x) + 0.25T_3(x)$$

at the point $x = 0.5$.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     C06DBF Example Program Text
*     Mark 14 Revised.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*     .. Local Scalars ..
      real             CALC, X
*     .. Local Arrays ..
      real             C(4)
*     .. External Functions ..
      real             C06DBF
      EXTERNAL         C06DBF
*     .. Data statements ..
      DATA             C/1.0e0, 1.0e0, 0.5e0, 0.25e0/
*     .. Executable Statements ..
      WRITE (NOUT,*) 'C06DBF Example Program Results'
      X = 0.5e0
      CALC = C06DBF(X,C,4,1)
      WRITE (NOUT,*)
      WRITE (NOUT,99999) 'Sum =', CALC
      STOP
*
99999 FORMAT (1X,A,F8.4)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
C06DBF Example Program Results

Sum =  0.5000
```